

# **utils**

## **Utility functions in GAP**

0.53

23 January 2018

**Sebastian Gutsche**

**Max Horn**

**Alexander Hulpke**

**Stefan Kohl**

**Frank Lübeck**

**Christopher D. Wensley**

**Sebastian Gutsche**

Email: [gutsche@mathematik.uni-kl.de](mailto:gutsche@mathematik.uni-kl.de)

Homepage: <http://wwwb.math.rwth-aachen.de/~gutsche/>

**Max Horn**

Email: [max.horn@math.uni-giessen.de](mailto:max.horn@math.uni-giessen.de)

Homepage: <http://www.quendi.de/math>

**Alexander Hulpke**

Email: [hulpke@math.colostate.edu](mailto:hulpke@math.colostate.edu)

Homepage: <http://www.math.colostate.edu/~hulpke>

**Stefan Kohl**

Email: [stefan@mcs.st-and.ac.uk](mailto:stefan@mcs.st-and.ac.uk)

Homepage: <https://www.gap-system.org/DevelopersPages/StefanKohl/>

**Frank Lübeck**

Email: [Frank.Luebeck@Math.RWTH-Aachen.De](mailto:Frank.Luebeck@Math.RWTH-Aachen.De)

Homepage: <http://www.math.rwth-aachen.de/~Frank.Luebeck>

**Christopher D. Wensley**

Email: [c.d.wensley@bangor.ac.uk](mailto:c.d.wensley@bangor.ac.uk)

Homepage: <http://pages.bangor.ac.uk/~mas023/>

## Abstract

The **Utils** package provides a space for utility functions in a variety of **GAP** packages to be collected together into a single package. In this way it is hoped that they will become more visible to package authors.

Any package author who transfers a function to **Utils** will become an author of **Utils**.

If deemed appropriate, functions may also be transferred from the main library.

Bug reports, suggestions and comments are, of course, welcome. Please contact the last author at [c.d.wensley@bangor.ac.uk](mailto:c.d.wensley@bangor.ac.uk) or submit an issue at the GitHub repository <https://github.com/gap-packages/utils/issues/>.

## Copyright

© 2015-2018, The GAP Group.

The **Utils** package is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or (at your option) any later version.

## Acknowledgements

This documentation was prepared with the **GAPDoc** [LN17] and **AutoDoc** [GH16] packages.

The procedure used to produce new releases uses the package **GitHubPagesForGAP** [Hor17] and the package **ReleaseTools**.

# Contents

<b>1</b>	<b>Introduction</b>	<b>4</b>
1.1	Information for package authors . . . . .	5
<b>2</b>	<b>Printing Lists and Iterators</b>	<b>6</b>
2.1	Printing selected items . . . . .	6
<b>3</b>	<b>Lists, Sets and Strings</b>	<b>8</b>
3.1	Functions for lists . . . . .	8
3.2	Distinct and Common Representatives . . . . .	10
3.3	Functions for strings . . . . .	11
<b>4</b>	<b>Number-theoretic functions</b>	<b>12</b>
4.1	Functions for integers . . . . .	12
<b>5</b>	<b>Groups and homomorphisms</b>	<b>15</b>
5.1	Functions for groups . . . . .	15
5.2	Functions for group homomorphisms . . . . .	17
<b>6</b>	<b>Records</b>	<b>18</b>
6.1	Functions for records . . . . .	18
<b>7</b>	<b>Various other functions</b>	<b>19</b>
7.1	Operations on folders . . . . .	19
7.2	File operations . . . . .	19
7.3	L <sup>A</sup> T <sub>E</sub> X strings . . . . .	20
7.4	Applicable methods . . . . .	20
7.5	Conversion to Magma string . . . . .	21
<b>8</b>	<b>The transfer procedure</b>	<b>23</b>
	<b>References</b>	<b>25</b>
	<b>Index</b>	<b>26</b>

# Chapter 1

## Introduction

The `Utils` package provides a space for utility functions from a variety of `GAP` packages to be collected together into a single package. In this way it is hoped that they will become more visible to other package authors. This package was first distributed as part of the `GAP` 4.8.2 distribution.

The package is loaded with the command

Example

```
gap> LoadPackage( "utils" );
```

Functions are currently being transferred from the following packages:

- Conversion of a `GAP` group to a Magma output string, taken from various sources including `other.gi` in the main library.

Transfer is complete (for now) for functions from the following packages:

- `AutoDoc` [GH16] (with function names changed);
- `ResClasses` [Koh17b];
- `RCWA` [Koh17a];
- `XMod` [WAOU17].

The package may be obtained either as a compressed `.tar` file or as a `.zip` file, `utils-version_number.tar.gz`, by ftp from one of the following sites:

- the `Utils` GitHub release site: <https://gap-packages.github.io/utils/>.
- any `GAP` archive, e.g. <https://www.gap-system.org/Packages/packages.html>;

The package also has a GitHub repository at: <https://github.com/gap-packages/utils>.

Once the package is loaded, the manual `doc/manual.pdf` can be found in the documentation folder. The `html` versions, with or without `MathJax`, may be rebuilt as follows:

Example

```
gap> ReadPackage( "utils", "makedoc.g" );
```

It is possible to check that the package has been installed correctly by running the test files:

Example

```
gap> ReadPackage( "utils", "tst/testing.g" );
#I Test output starts here
...
#I No errors detected while testing package utils
true
```

Note that functions listed in this manual that are currently in the process of being transferred are only read from the source package **Home** (say), and so can only be used if **Home** has already been loaded.

## 1.1 Information for package authors

A function (or collection of functions) is suitable for transfer from a package **Home** to **Utils** if the following conditions are satisfied.

- The function is sufficiently non-specialised so that it might be of use to other authors.
- The function does not depend on the remaining functions in **Home**
- The function does not do what can already be done with a **GAP** library function.
- Documentation of the function and test examples are available.
- When there is more than one active author of **Home**, they should all be aware (and content) that the transfer is taking place.

Authors of packages may be reluctant to let go of their utility functions. The following principles may help to reassure them. (Suggestions for more items here are welcome.)

- A function that has been transferred to **Utils** will not be changed without the approval of the original author.
- The current package maintainer has every intention of continuing to maintain **Utils**. In the event that this proves impossible, the **GAP** development team will surely find someone to take over.
- Function names will not be changed unless specifically requested by **Home**'s author(s) or unless they have the form `HOME_FunctionName`.
- In order to speed up the transfer process, only functions from one package will be in transition at any given time. Hopefully a week or two will suffice for most packages.
- Any package author who transfers a function to **Utils** will become an author of **Utils**. (In truth, **Utils** does not have *authors*, just a large number of *contributors*.)

The process for transferring utility functions from **Home** to **Utils** is described in Chapter 8.

## Chapter 2

# Printing Lists and Iterators

### 2.1 Printing selected items

The functions described here print lists or objects with an iterator with one item per line, either the whole list/iterator or certain subsets:

- by giving a list of positions of items to be printed, or
- by specifying a first item and then a regular step.

#### 2.1.1 PrintOneItemPerLine

▷ `PrintOneItemPerLine(obj)`

(function)

This function calls operations `PrintListOneItemPerLine` (which has been transferred from package `XMod`) or `PrintIteratorOneItemPerLine`.

Printing lists vertically, rather than horizontally, may be useful when the entries are lengthy. This function does this for lists, iterators, and objects which have an iterator.

Example

```
gap> s3 := SymmetricGroup( 3 );;
gap> L := KnownPropertiesOfObject( GeneratorsOfGroup( s3 ) );;
gap> PrintOneItemPerLine( L );
[ IsFinite,
  IsSmallList,
  IsGeneratorsOfMagmaWithInverses,
  IsGeneratorsOfSemigroup,
  IsSubsetLocallyFiniteGroup
]
gap> PrintOneItemPerLine( s3 );
()
(2,3)
(1,3)
(1,3,2)
(1,2,3)
(1,2)
```

### 2.1.2 PrintSelection

- ▷ `PrintSelection(obj, first, step[, last])` (function)
- ▷ `PrintSelection(obj, list)` (function)

This function, given three (or four) parameters, calls operations `PrintSelectionFromList` or `PrintSelectionFromIterator` which prints the *first* item specified, and then the item at every *step*. The fourth parameter is essential when the object being printed is infinite.

Alternatively, given two parameters, with the second parameter a list *L* of positive integers, only the items at positions in *L* are printed.

Example

```
gap> L := List( [1..20], n -> n^5 );
gap> PrintSelection( L, 1, 8 );
1 : 1
9 : 59049
17 : 1419857
gap> PrintSelection( L, [9..11] );
9 : 59049
10 : 100000
11 : 161051
gap> s5 := SymmetricGroup( 5 );
gap> PrintSelection( s5, 12, 30 );
12 : (2,5)(3,4)
42 : (1,5,2,3,4)
72 : (1,2,4,3)
102 : (1,4,3,5)
gap> PrintSelection( s5, [118..122] );
118 : (1,4,3,2,5)
119 : (1,4,2,5,3)
120 : (1,4,3)(2,5)
gap> f2 := FreeGroup( 2 );
gap> PrintSelection( f2, 20, 20, 100 );
20 : f2*f1^2
40 : f2*f1*f2
60 : f1*f2*f1^2
80 : (f2*f1)^2
100 : f1*f2^2*f1^-1
gap> PrintSelection( f2, [11..15] );
11 : f2^-1*f1^-1
12 : f1*f2
13 : f1^-1*f2
14 : f1*f2^-1
15 : f1^-1*f2^-1
```



## Chapter 3

# Lists, Sets and Strings

### 3.1 Functions for lists

#### 3.1.1 DifferencesList

▷ DifferencesList( $L$ ) (function)

This function has been transferred from package **ResClasses**.

It takes a list  $L$  of length  $n$  and outputs the list of length  $n - 1$  containing all the differences  $L[i] - L[i - 1]$ .

Example

```
gap> List( [1..12], n->n^3 );
[ 1, 8, 27, 64, 125, 216, 343, 512, 729, 1000, 1331, 1728 ]
gap> DifferencesList( last );
[ 7, 19, 37, 61, 91, 127, 169, 217, 271, 331, 397 ]
gap> DifferencesList( last );
[ 12, 18, 24, 30, 36, 42, 48, 54, 60, 66 ]
gap> DifferencesList( last );
[ 6, 6, 6, 6, 6, 6, 6, 6, 6 ]
```

#### 3.1.2 QuotientsList

▷ QuotientsList( $L$ ) (function)

▷ FloatQuotientsList( $L$ ) (function)

These functions have been transferred from package **ResClasses**.

They take a list  $L$  of length  $n$  and output the quotients  $L[i]/L[i - 1]$  of consecutive entries in  $L$ . An error is returned if an entry is zero.

Example

```
gap> List( [0..10], n -> Factorial(n) );
[ 1, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880, 3628800 ]
gap> QuotientsList( last );
[ 1, 2, 3, 4, 5, 6, 7, 8, 9, 10 ]
```

```

gap> L := [ 1, 3, 5, -1, -3, -5 ];;
gap> QuotientsList( L );
[ 3, 5/3, -1/5, 3, 5/3 ]
gap> FloatQuotientsList( L );
[ 3., 1.66667, -0.2, 3., 1.66667 ]
gap> QuotientsList( [ 2, 1, 0, -1, -2 ] );
[ 1/2, 0, fail, 2 ]
gap> FloatQuotientsList( [1..10] );
[ 2., 1.5, 1.33333, 1.25, 1.2, 1.16667, 1.14286, 1.125, 1.11111 ]
gap> Product( last );
10.

```

### 3.1.3 SearchCycle

▷ SearchCycle(L)

(operation)

This function has been transferred from package RCWA.

SearchCycle is a tool to find likely cycles in lists. What, precisely, a *cycle* is, is deliberately fuzzy here, and may possibly even change. The idea is that the beginning of the list may be anything, following that the same pattern needs to be repeated several times in order to be recognized as a cycle.

Example

```

gap> L := [1..20];; L[1]:=13;;
gap> for i in [1..19] do
>   if IsOddInt(L[i]) then L[i+1]:=3*L[i]+1; else L[i+1]:=L[i]/2; fi;
>   od;
gap> L;
[ 13, 40, 20, 10, 5, 16, 8, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4 ]
gap> SearchCycle( L );
[ 1, 4, 2 ]
gap> n := 1;; L := [n];;
gap> for i in [1..100] do n:=(n^2+1) mod 1093; Add(L,n); od;
gap> L;
[ 1, 2, 5, 26, 677, 363, 610, 481, 739, 715, 795, 272, 754, 157, 604, 848,
  1004, 271, 211, 802, 521, 378, 795, 272, 754, 157, 604, 848, 1004, 271,
  211, 802, 521, 378, 795, 272, 754, 157, 604, 848, 1004, 271, 211, 802, 521,
  378, 795, 272, 754, 157, 604, 848, 1004, 271, 211, 802, 521, 378, 795, 272,
  754, 157, 604, 848, 1004, 271, 211, 802, 521, 378, 795, 272, 754, 157, 604,
  848, 1004, 271, 211, 802, 521, 378, 795, 272, 754, 157, 604, 848, 1004,
  271, 211, 802, 521, 378, 795, 272, 754, 157, 604, 848, 1004 ]
gap> C := SearchCycle( L );
[ 157, 604, 848, 1004, 271, 211, 802, 521, 378, 795, 272, 754 ]
gap> P := Positions( L, 157 );
[ 14, 26, 38, 50, 62, 74, 86, 98 ]
gap> Length( C ); DifferencesList( P );
12
[ 12, 12, 12, 12, 12, 12, 12 ]

```

### 3.1.4 RandomCombination

▷ `RandomCombination( $S$ ,  $k$ )` (operation)

This function has been transferred from package `ResClasses`.  
It returns a random unordered  $k$ -tuple of distinct elements of a set  $S$ .

Example

```
gap> ## "6 aus 49" is a common lottery in Germany
gap> RandomCombination( [1..49], 6 );
[ 2, 16, 24, 26, 37, 47 ]
```

## 3.2 Distinct and Common Representatives

### 3.2.1 DistinctRepresentatives

▷ `DistinctRepresentatives( $list$ )` (operation)  
 ▷ `CommonRepresentatives( $list$ )` (operation)  
 ▷ `CommonTransversal( $grp$ ,  $subgrp$ )` (operation)  
 ▷ `IsCommonTransversal( $grp$ ,  $subgrp$ ,  $list$ )` (operation)

These functions have been transferred from package `XMod`.

They deal with lists of subsets of  $[1 \dots n]$  and construct systems of distinct and common representatives using simple, non-recursive, combinatorial algorithms.

When  $L$  is a set of  $n$  subsets of  $[1 \dots n]$  and the Hall condition is satisfied (the union of any  $k$  subsets has at least  $k$  elements), a set of `DistinctRepresentatives` exists.

When  $J, K$  are both lists of  $n$  sets, the function `CommonRepresentatives` returns two lists: the set of representatives, and a permutation of the subsets of the second list. It may also be used to provide a common transversal for sets of left and right cosets of a subgroup  $H$  of a group  $G$ , although a greedy algorithm is usually quicker.

Example

```
gap> J := [ [1,2,3], [3,4], [3,4], [1,2,4] ];;
gap> DistinctRepresentatives( J );
[ 1, 3, 4, 2 ]
gap> K := [ [3,4], [1,2], [2,3], [2,3,4] ];;
gap> CommonRepresentatives( J, K );
[ [ 3, 3, 3, 1 ], [ 1, 3, 4, 2 ] ]
gap> d16 := DihedralGroup( IsPermGroup, 16 );
Group([ (1,2,3,4,5,6,7,8), (2,8)(3,7)(4,6) ])
gap> SetName( d16, "d16" );
gap> c4 := Subgroup( d16, [ d16.1^2 ] );
Group([ (1,3,5,7)(2,4,6,8) ])
gap> SetName( c4, "c4" );
gap> RightCosets( d16, c4 );
[ RightCoset(c4,()), RightCoset(c4,(2,8)(3,7)(4,6)), RightCoset(c4,(1,8,7,6,5,
  4,3,2)), RightCoset(c4,(1,8)(2,7)(3,6)(4,5)) ]
gap> trans := CommonTransversal( d16, c4 );
```

```
[ (), (2,8)(3,7)(4,6), (1,2,3,4,5,6,7,8), (1,2)(3,8)(4,7)(5,6) ]
gap> IsCommonTransversal( d16, c4, trans );
true
```

### 3.3 Functions for strings

#### 3.3.1 BlankFreeString

▷ BlankFreeString(obj) (function)

This function has been transferred from package ResClasses.

The result of BlankFreeString( obj ); is a composite of the functions String( obj ) and RemoveCharacters( obj, " " );.

Example

```
gap> gens := GeneratorsOfGroup( DihedralGroup(12) );
[ f1, f2, f3 ]
gap> String( gens );
"[ f1, f2, f3 ]"
gap> BlankFreeString( gens );
"[f1,f2,f3]"
```

#### 3.3.2 StringDotSuffix

▷ StringDotSuffix(str, suf) (operation)

This function has been transferred from package AutoDoc, where it was originally named AUTODOC\_GetSuffix.

When StringDotSuffix is given a string containing a "." it return its extension, i.e. the bit after the last ".".

Example

```
gap> StringDotSuffix( "file.ext" );
"ext"
gap> StringDotSuffix( "file.ext.bak" );
"bak"
gap> StringDotSuffix( "file." );
""
gap> StringDotSuffix( "Hello" );
fail
```

## Chapter 4

# Number-theoretic functions

### 4.1 Functions for integers

#### 4.1.1 AllSmoothIntegers

- ▷ AllSmoothIntegers(*maxp*, *maxn*) (function)
- ▷ AllSmoothIntegers(*maxp*, *L*) (function)

This function has been transferred from package RCWA.

The function AllSmoothIntegers(*maxp*, *maxn*) returns the list of all positive integers less than or equal to *maxn* whose prime factors are all in the list  $L = \{p \mid p \leq \text{maxp}, p \text{ prime}\}$ .

In the alternative form, when *L* is a list of primes, the function returns the list of all positive integers whose prime factors lie in *L*.

Example

```
gap> AllSmoothIntegers( 3, 1000 );
[ 1, 2, 3, 4, 6, 8, 9, 12, 16, 18, 24, 27, 32, 36, 48, 54, 64, 72, 81, 96,
  108, 128, 144, 162, 192, 216, 243, 256, 288, 324, 384, 432, 486, 512, 576,
  648, 729, 768, 864, 972 ]
gap> AllSmoothIntegers( [5,11,17], 1000 );
[ 1, 5, 11, 17, 25, 55, 85, 121, 125, 187, 275, 289, 425, 605, 625, 935 ]
gap> Length( last );
16
gap> List( [3..20], n -> Length( AllSmoothIntegers( [5,11,17], 10^n ) ) );
[ 16, 29, 50, 78, 114, 155, 212, 282, 359, 452, 565, 691, 831, 992, 1173,
  1374, 1595, 1843 ]
```

#### 4.1.2 AllProducts

- ▷ AllProducts(*L*, *k*) (function)

This function has been transferred from package RCWA.

The command AllProducts(*L*, *k*) returns the list of all products of *k* entries of the list *L*. Note that every ordering of the entries is used so that, in the commuting case, there are bound to be repetitions.

## Example

```
gap> AllProducts([1..4],3);
[ 1, 2, 3, 4, 2, 4, 6, 8, 3, 6, 9, 12, 4, 8, 12, 16, 2, 4, 6, 8, 4, 8, 12,
  16, 6, 12, 18, 24, 8, 16, 24, 32, 3, 6, 9, 12, 6, 12, 18, 24, 9, 18, 27,
  36, 12, 24, 36, 48, 4, 8, 12, 16, 8, 16, 24, 32, 12, 24, 36, 48, 16, 32,
  48, 64 ]
gap> Set(last);
[ 1, 2, 3, 4, 6, 8, 9, 12, 16, 18, 24, 27, 32, 36, 48, 64 ]
gap> AllProducts( [(1,2,3),(2,3,4)], 2 );
[ (2,4,3), (1,2)(3,4), (1,3)(2,4), (1,3,2) ]
```

### 4.1.3 RestrictedPartitionsWithoutRepetitions

▷ RestrictedPartitionsWithoutRepetitions( $n$ ,  $S$ )

(function)

This function has been transferred from package RCWA.

For a positive integer  $n$  and a set of positive integers  $S$ , this function returns the list of partitions of  $n$  into distinct elements of  $S$ . Unlike RestrictedPartitions, no repetitions are allowed.

## Example

```
gap> RestrictedPartitions( 20, [4..10] );
[ [ 4, 4, 4, 4, 4 ], [ 5, 5, 5, 5 ], [ 6, 5, 5, 4 ], [ 6, 6, 4, 4 ],
  [ 7, 5, 4, 4 ], [ 7, 7, 6 ], [ 8, 4, 4, 4 ], [ 8, 6, 6 ], [ 8, 7, 5 ],
  [ 8, 8, 4 ], [ 9, 6, 5 ], [ 9, 7, 4 ], [ 10, 5, 5 ], [ 10, 6, 4 ],
  [ 10, 10 ] ]
gap> RestrictedPartitionsWithoutRepetitions( 20, [4..10] );
[ [ 10, 6, 4 ], [ 9, 7, 4 ], [ 9, 6, 5 ], [ 8, 7, 5 ] ]
gap> RestrictedPartitionsWithoutRepetitions( 10^2, List([1..10], n->n^2 ) );
[ [ 100 ], [ 64, 36 ], [ 49, 25, 16, 9, 1 ] ]
```

### 4.1.4 ExponentOfPrime

▷ ExponentOfPrime( $n$ ,  $p$ )

(function)

This function has been transferred from package RCWA.

ExponentOfPrime( $n$ ,  $p$ ) returns the exponent of the prime  $p$  in the prime factorization of  $n$ .

## Example

```
gap> ExponentOfPrime( 13577531, 11 );
3
gap> List( [1..40], n -> ExponentOfPrime( 3^n-1, 2 ) );
[ 1, 3, 1, 4, 1, 3, 1, 5, 1, 3, 1, 4, 1, 3, 1, 6, 1, 3, 1, 4, 1, 3, 1, 5, 1,
  3, 1, 4, 1, 3, 1, 7, 1, 3, 1, 4, 1, 3, 1, 5 ]
gap> List( [1..40], n -> ExponentOfPrime( n^2-1, 2 ) );
[ infinity, 0, 3, 0, 3, 0, 4, 0, 4, 0, 3, 0, 3, 0, 5, 0, 5, 0, 3, 0, 3, 0, 4,
  0, 4, 0, 3, 0, 3, 0, 6, 0, 6, 0, 3, 0, 3, 0, 4, 0 ]
```

### 4.1.5 NextProbablyPrimeInt

▷ NextProbablyPrimeInt(*n*)

(function)

This function has been transferred from package RCWA.

The function NextProbablyPrimeInt(*n*) does the same as NextPrimeInt(*n*) except that for reasons of performance it tests numbers only for IsProbablyPrimeInt(*n*) instead of IsPrimeInt(*n*). For large *n*, this function is much faster than NextPrimeInt(*n*)

Example

```
gap> n := 2^251;
3618502788666131106986593281521497120414687020801267626233049500247285301248
gap> time;
0
gap> NextProbablyPrimeInt( n );
3618502788666131106986593281521497120414687020801267626233049500247285301313
gap> time;
1
gap> NextPrimeInt( n );
3618502788666131106986593281521497120414687020801267626233049500247285301313
gap> time;
12346
```

### 4.1.6 PrimeNumbersIterator

▷ PrimeNumbersIterator([*chunksize*])

(function)

This function has been transferred from package RCWA.

This function returns an iterator which runs over the prime numbers *n* ascending order; it takes an optional argument *chunksize* which specifies the length of the interval which is sieved in one go (the default is  $10^7$ ), and which can be used to balance runtime vs. memory consumption. It is assumed that *chunksize* is larger than any gap between two consecutive primes within the range one intends to run the iterator over.

Example

```
gap> iter := PrimeNumbersIterator();;
gap> for i in [1..100] do p := NextIterator(iter); od;
gap> p;
541
gap> sum := 0;;
gap> ## "prime number race" 1 vs. 3 mod 4
gap> for p in PrimeNumbersIterator() do
>   if p <> 2 then sum := sum + E(4)^(p-1); fi;
>   if sum > 0 then break; fi;
> od;
gap> p;
26861
```

## Chapter 5

# Groups and homomorphisms

### 5.1 Functions for groups

#### 5.1.1 Comm

▷ `Comm(L)` (operation)

This method has been transferred from package `ResClasses`.

It provides a method for `Comm` when the argument is a list (enclosed in square brackets), and calls the function `LeftNormedComm`.

Example

```
gap> Comm( [ (1,2), (2,3) ] );
(1,2,3)
gap> Comm( [(1,2),(2,3),(3,4),(4,5),(5,6)] );
(1,5,6)
gap> Comm(Comm(Comm(Comm((1,2),(2,3)),(3,4)),(4,5)),(5,6)); ## the same
(1,5,6)
```

#### 5.1.2 IsCommuting

▷ `IsCommuting(a, b)` (operation)

This function has been transferred from package `ResClasses`.

It tests whether two elements in a group commute.

Example

```
gap> D12 := DihedralGroup( 12 );
<pc group of size 12 with 3 generators>
gap> SetName( D12, "D12" );
gap> a := D12.1;; b := D12.2;;
gap> IsCommuting( a, b );
false
```



### 5.1.3 ListOfPowers

▷ `ListOfPowers( $g$ ,  $exp$ )` (operation)

This function has been transferred from package **RCWA**.

The operation `ListOfPowers( $g$ ,  $exp$ )` returns the list  $[g, g^2, \dots, g^{exp}]$  of powers of the element  $g$ .

Example

```
gap> ListOfPowers( 2, 20 );
[ 2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048, 4096, 8192, 16384,
  32768, 65536, 131072, 262144, 524288, 1048576 ]
gap> ListOfPowers( (1,2,3)(4,5), 12 );
[ (1,2,3)(4,5), (1,3,2), (4,5), (1,2,3), (1,3,2)(4,5), (),
  (1,2,3)(4,5), (1,3,2), (4,5), (1,2,3), (1,3,2)(4,5), () ]
gap> ListOfPowers( D12.2, 6 );
[ f2, f3, f2*f3, f3^2, f2*f3^2, <identity> of ... ]
```

### 5.1.4 GeneratorsAndInverses

▷ `GeneratorsAndInverses( $G$ )` (operation)

This function has been transferred from package **RCWA**.

This operation returns a list containing the generators of  $G$  followed by the inverses of these generators.

Example

```
gap> GeneratorsAndInverses( D12 );
[ f1, f2, f3, f1, f2*f3^2, f3^2 ]
gap> GeneratorsAndInverses( SymmetricGroup(5) );
[ (1,2,3,4,5), (1,2), (1,5,4,3,2), (1,2) ]
```

### 5.1.5 UpperFittingSeries

▷ `UpperFittingSeries( $G$ )` (attribute)

▷ `LowerFittingSeries( $G$ )` (attribute)

▷ `FittingLength( $G$ )` (attribute)

These three functions have been transferred from package **ResClasses**.

The upper and lower Fitting series and the Fitting length of a solvable group are described here:

[https://en.wikipedia.org/wiki/Fitting\\_length](https://en.wikipedia.org/wiki/Fitting_length).

Example

```
gap> UpperFittingSeries( D12 ); LowerFittingSeries( D12 );
[ Group([ ]), Group([ f3, f2*f3 ]), Group([ f3, f2*f3, f1 ]) ]
[ D12, Group([ f3 ]), Group([ ]) ]
gap> FittingLength( D12 );
2
```

```

gap> S4 := SymmetricGroup( 4 );;
gap> UpperFittingSeries( S4 );
[ Group(), Group([ (1,2)(3,4), (1,4)(2,3) ]), Group([ (1,2)(3,4), (1,4)
  (2,3), (2,4,3) ]), Group([ (3,4), (2,3,4), (1,2)(3,4) ]) ]
gap> List( last, StructureDescription );
[ "1", "C2 x C2", "A4", "S4" ]
gap> LowerFittingSeries( S4 );
[ Sym( [ 1 .. 4 ] ), Alt( [ 1 .. 4 ] ), Group([ (1,4)(2,3), (1,3)
  (2,4) ]), Group() ]
gap> List( last, StructureDescription );
[ "S4", "A4", "C2 x C2", "1" ]
gap> FittingLength( S4);
3

```

## 5.2 Functions for group homomorphisms

### 5.2.1 EpimorphismByGenerators

▷ `EpimorphismByGenerators( $G$ ,  $H$ )` (operation)

This function has been transferred from package RCWA.

It constructs a group homomorphism which maps the generators of  $G$  to those of  $H$ . Its intended use is when  $G$  is a free group, and a warning is printed when this is not the case. Note that anything may happen if the resulting map is not a homomorphism!

Example

```

gap> G := Group( (1,2,3), (3,4,5), (5,6,7), (7,8,9) );;
gap> phi := EpimorphismByGenerators( FreeGroup("a","b","c","d"), G );
[ a, b, c, d ] -> [ (1,2,3), (3,4,5), (5,6,7), (7,8,9) ]
gap> PreImagesRepresentative( phi, (1,2,3,4,5,6,7,8,9) );
d*c*b*a
gap> a := G.1;; b := G.2;; c := G.3;; d := G.4;;
gap> d*c*b*a;
(1,2,3,4,5,6,7,8,9)
gap> ## note that it is easy to produce nonsense:
gap> epi := EpimorphismByGenerators( Group((1,2,3)), Group((8,9)) );
Warning: calling GroupHomomorphismByImagesNC without checks
[ (1,2,3) ] -> [ (8,9) ]
gap> IsGroupHomomorphism( epi );
true
gap> Image( epi, (1,2,3) );
()
gap> Image( epi, (1,3,2) );
(8,9)

```

## Chapter 6

# Records

### 6.1 Functions for records

#### 6.1.1 SetIfMissing

▷ `SetIfMissing(rec, name, val)` (function)

This function has been transferred from package `AutoDoc`, where it was called `AUTODOC_WriteOnce`. It writes into a record provided the position is not yet bound.

Example

```
gap> r := rec( a := 1, b := 2 );;
gap> SetIfMissing( r, "c", 3 );
gap> names := RecNames( r );;
gap> Set( names );
[ "a", "b", "c" ]
gap> SetIfMissing( r, "c", 4 );
gap> r;
rec( a := 1, b := 2, c := 3 )
```

#### 6.1.2 AssignGlobals

▷ `AssignGlobals(rec)` (function)

This function has been transferred from package `RCWA`.

It assigns the record components of `rec` to global variables with the same names.

Example

```
gap> AssignGlobals( r );
The following global variables have been assigned:
[ "a", "b", "c" ]
gap> [a,b,c];
[ 1, 2, 3 ]
```

## Chapter 7

# Various other functions

### 7.1 Operations on folders

#### 7.1.1 FindMatchingFiles

- ▷ FindMatchingFiles(*pkg*, *dirs*, *extns*) (function)
- ▷ CreateDirIfMissing(*str*) (function)

These functions have been transferred from package `AutoDoc` where they were named `AutoDoc_FindMatchingFiles` and `AutoDoc_CreateDirIfMissing`.

`FindMatchingFiles` scans the given (by name) subdirectories of a package directory for files with one of the given extensions, and returns the corresponding filenames, as paths relative to the package directory.

`CreateDirIfMissing` checks whether the given directory exists and, if not, attempts to create it. In either case `true` is returned.

*Warning:* this function relies on the undocumented library function `CreateDir`, so use it with caution.

Example

```
gap> FindMatchingFiles( "utils", [ "/", "tst" ], [ "g", "txt" ] );
[ "/LICENSE.txt", "/PackageInfo.g", "/init.g", "/makedoc.g", "/read.g",
  "tst/testall.g" ]
gap> CreateDirIfMissing( "/Applications/gap/temp/" );
true
```

### 7.2 File operations

#### 7.2.1 Log2HTML

- ▷ Log2HTML(*filename*) (function)

This function has been transferred from package `RCWA`.

This function converts the **GAP** logfile *filename* to HTML. The extension of the input file must be `*.log`. The name of the output file is the same as the one of the input file except that the extension

`*.log` is replaced by `*.html`. There is a sample CSS file in `utils/doc/gaplog.css`, which you can adjust to your taste.

Example

```
gap> LogTo("mar2.log");
gap> FindMatchingFiles( "utils", [""], ["g"] );
[ "/PackageInfo.g", "/init.g", "/makedoc.g", "/read.g" ]
gap> LogTo();
gap> Log2HTML( "mar2.log" );
gap> FindMatchingFiles( "utils", [""], ["html", "log"] );
[ "/mar2.html", "/mar2.log" ]
```

## 7.3 $\text{\LaTeX}$ strings

### 7.3.1 IntOrOnfinityToLaTeX

▷ `IntOrOnfinityToLaTeX( $n$ )` (function)

This function has been transferred from package `ResClasses`.

`IntOrInfinityToLaTeX( $n$ )` returns the  $\text{\LaTeX}$  string for  $n$ .

Example

```
gap> IntOrInfinityToLaTeX( 10^3 );
"1000"
gap> IntOrInfinityToLaTeX( infinity );
"\\infty"
```

### 7.3.2 LaTeXStringFactorsInt

▷ `LaTeXStringFactorsInt( $n$ )` (function)

This function has been transferred from package `RCWA`.

It returns the prime factorization of the integer  $n$  as a string in  $\text{\LaTeX}$  format.

Example

```
gap> LaTeXStringFactorsInt( Factorial(12) );
"2^{10} \\cdot 3^5 \\cdot 5^2 \\cdot 7 \\cdot 11"
```

## 7.4 Applicable methods

### 7.4.1 PrintApplicableMethod

▷ `PrintApplicableMethod( $arg$ )` (function)

This function combines calls to `ApplicableMethod`, `FilenameFunc`, `StartlineFunc` and `EndlineFunc` and prints the location of the file containing the method found, and a listing of that method. In its simplest form it is called as `PrintApplicableMethod(f,L)` for a function `f` and a list of parameters `L`. Alternatively, it is called as `PrintApplicableMethod(f,L,0,n)` and then prints the method returned by `ApplicableMethod(f,L,0,n)`.

Example

```
gap> PrintApplicableMethod( IsCyclic, [ Group((1,2,3),(4,5)) ] );
this method is contained in lines [30,36] of file:
/Applications/gap/gapdev/lib/grp.gi
function ( G )
  if Length( GeneratorsOfGroup( G ) ) = 1 then
    return true;
  else
    return TRY_NEXT_METHOD;
  fi;
  return;
end
gap> PrintApplicableMethod( IsCyclic, [ Group((1,2,3),(4,5)) ], 0, 2 );
this method is contained in lines [41,63] of file:
/Applications/gap/gapdev/lib/grp.gi
function ( G )
  if HasGeneratorsOfGroup( G ) and Length( GeneratorsOfGroup( G ) ) = 1
    then
      SetMinimalGeneratingSet( G, GeneratorsOfGroup( G ) );
      return true;
    elif not IsCommutative( G ) then
      return false;
    elif IsFinite( G ) then
      return ForAll( Set( FactorsInt( Size( G ) ) ), function ( p )
        return
          Index( G,
            SubgroupNC( G,
              List( GeneratorsOfGroup( G ), function ( g )
                return g ^ p;
              end ) ) ) = p;
          end );
    else
      return AbelianInvariants( G ) = [ 0 ];
    fi;
  return;
end
```

## 7.5 Conversion to Magma string

### 7.5.1 ConvertToMagmaInputString

▷ `ConvertToMagmaInputString(arg)`

(function)

The function `ConvertToMagmaInputString( obj [, str] )` attempts to output a string `s` which can be read into Magma [BCP97] so as to produce the same group in that computer algebra system. In the second form the user specifies the name of the resulting object, so that the output string has the form `"str := ..."`. When `obj` is a permutation group, the operation `PermGroupToMagmaFormat(obj)` is called. This function has been taken from `other.gi` in the main library where it was called `MagmaInputString`. When `obj` is a pc-group, the operation `PcGroupToMagmaFormat(obj)` is called. This function was private code of Max Horn. When `obj` is a matrix group over a finite field, the operation `MatrixGroupToMagmaFormat(obj)` is called. This function is a modification of private code of Frank Lübeck.

Hopefully code for other types of group will be added in due course.

These functions should be considered *experimental*, and more testing is desirable.

— Example —

```
gap> ConvertToMagmaInputString( Group( (1,2,3,4,5), (3,4,5) ) );
"PermutationGroup<5|(1,2,3,4,5),\n(3,4,5)>;\n"
gap> ConvertToMagmaInputString( Group( (1,2,3,4,5) ), "c5" );
"c5:=PermutationGroup<5|(1,2,3,4,5)>;\n"
gap> ConvertToMagmaInputString( SmallGroup( 24, 12 ) );
"PolycyclicGroup< f1,f2,f3,f4 |\nf1^2,\nf2^3,\nf3^2,\nf4^2,\nf2^f1 = f2^2,\nf3\
^f1 = f4,\nf3^f2 = f4,\nf4^f1 = f3,\nf4^f2 = f3*f4\n>;\n"
gap> ConvertToMagmaInputString( CyclicGroup( IsPcGroup, 7 ), "c7" );
"c7:=PolycyclicGroup< f1 |\nf1^7\n>;\n"
gap> M := GL(2,5);; Size(M);
480
gap> s1 := ConvertToMagmaInputString( M );
"F := GF(5);\nP := GL(2,F);\ngens := [\nP![2,0,0,1],\nP![4,1,4,0]\n];\nsub<P | \
gens>;\n"
gap> Print( s1 );
F := GF(5);
P := GL(2,F);
gens := [
P![2,0,0,1],
P![4,1,4,0]
];
sub<P | gens>;
gap> n1 := [ [ Z(9)^0, Z(9)^0 ], [ Z(9)^0, Z(9) ] ];;
gap> n2 := [ [ Z(9)^0, Z(9)^3 ], [ Z(9)^4, Z(9)^2 ] ];;
gap> N := Group( n1, n2 );; Size( N );
5760
gap> s2 := ConvertToMagmaInputString( N, "gpN" );;
gap> Print( s2 );
F := GF(3^2);
P := GL(2,F);
w := PrimitiveElement(F);
gens := [
P![ 1, 1, 1,w^1],
P![ 1,w^3, 2,w^2]
];
gpN := sub<P | gens>;
```

## Chapter 8

# The transfer procedure

We consider here the process for transferring utility functions from a package **Home** to **Utils** which has to avoid the potential problem of duplicate declarations of a function causing loading problems in **GAP**.

If the functions in **Home** all have names of the form `HOME_FunctionName` then, in **Utils**, these functions are likely to be renamed as `FunctionName` or something similar. In this case the problem of duplicate declarations does not arise. This is what has happened with transfers from the **AutoDoc** package.

The case where the function names are unchanged is more complicated. Initially we tried out a process which allowed repeated declarations and installations of the functions being transferred. This involved additions to the main library files `global.g` and `oper.g`. Since there were misgivings about interfering in this way with basic operations such as `BIND_GLOBAL`, a simpler (but slightly less convenient) process has been adopted.

Using this alternative procedure, the following steps will be followed when making transfers from **Home** to **Utils**.

1. (**Home**:) Offer functions for inclusion. This may be simply done by emailing a list of functions. More usefully, email the declaration, implementation, test and documentation files, e.g.: `home.gd`, `home.gi`, `home.tst` and `home.xml`. (All active authors should be involved.)
2. (**Home**:) Declare that `M.N` is the last version of **Home** to contain these functions, so that `M.N+1` (or similar) will be the first version of **Home** to have all these functions removed, and to specify **Utils** as a required package.
3. (**Utils**:) Add strings `"home"` and `"m.n"` to the list `UtilsPackageVersions` in the file `utils/lib/start.gd`.

Example

```
UtilsPackageVersions :=  
  [ "autodoc",    "2016.01.31",  
    "resclasses", "4.2.5",  
    "home",       "m.n",  
    ...,          ...  
  ];
```



While the transfers are being made, it is essential that any new versions of **Home** should be tested with the latest version of **Utils** before they are released, so as to avoid loading failures.

4. (**Utils**;) Include the function declaration and implementation sections in suitable files, enclosed within a conditional clause of the form:

Example

```
if OKtoReadFromUtils( "Home" ) then
. . . . .
  <the code>
. . . . .
fi;
```

The function `OKtoReadFromUtils` returns `true` only if there is an installed version of **Home** and if this version is greater than M.N. So, at this stage, *the copied code will not be read*, and the transferred functions can only be called if **Home** has been installed.

5. (**Utils**;) Add the test and documentation material to the appropriate files. The copied code can be tested by temporarily moving **Home** away from **GAP**'s package directory.
6. (**Utils**;) Release a new version of **Utils** containing all the transferred material.
7. (**Home**;) Edit out the declarations and implementations of all the transferred functions, and remove references to them in the manual and tests. Possibly add a note to the manual that these functions have been transferred. Add **Utils** to the list of **Home**'s required packages in `PackageInfo.g`. Release a new version of **Home**.
8. (**Utils**;) In due course, when the new version(s) of **Home** are well established, it may be safe to remove the conditional clauses mentioned in item 4 above. The entry for **Home** in `UtilsPackageLists` may then be removed.

Finally, a note on the procedure for testing these functions. As long as a function being transferred still exists in the **Home** package, the code will not be read from **Utils**. So, when the tests are run, it is necessary to `LoadPackage("home")` before the function is called. The file `utils/tst/testall.g` makes sure that all the necessary packages are loaded before the individual tests are called.

# References

- [BCP97] W. Bosma, J. Cannon, and C. Playoust. *The Magma algebra system. {I}. The user language*, 1997. Computational algebra and number theory (London, 1993)} <http://dx.doi.org/10.1006/jscs.1996.0125>. 22
- [GH16] S. Gutsche and M. Horn. *AutoDoc - Generate documentation from GAP source code (Version 2016.12.04)*, 2016. GAP package, <https://github.com/gap-packages/AutoDoc>. 2, 4
- [Hor17] M. Horn. *GitHubPagesForGAP - Template for easily using GitHub Pages within GAP packages (Version 0.2)*, 2017. GAP package, <https://gap-system.github.io/GitHubPagesForGAP/>. 2
- [Koh17a] S. Kohl. *RCWA - Residue-Class-Wise Affine Groups (Version 4.5.1)*, 2017. GAP package, <https://stefan-kohl.github.io/rcwa.html>. 4
- [Koh17b] S. Kohl. *ResClasses - Set-Theoretic Computations with Residue Classes (Version 4.6.0)*, 2017. GAP package, <https://stefan-kohl.github.io/resclasses.html>. 4
- [LN17] F. Lübeck and M. Neunhöffer. *GAPDoc (Version 1.6)*. RWTH Aachen, 2017. GAP package, <http://www.math.rwth-aachen.de/~Frank.Luebeck/GAPDoc/index.html>. 2
- [WAOU17] C. D. Wensley, M. Alp, A. Odabas, and E. O. Uslu. *XMod - Crossed Modules and Cat1-groups in GAP (Version 2.64)*, 2017. GAP package, <https://github.com/gap-packages/xmod>. 4

# Index

AllProducts, 12  
AllSmoothIntegers, 12  
AssignGlobals, 18  
BlankFreeString, 11  
Comm, 15  
CommonRepresentatives, 10  
CommonTransversal, 10  
ConvertToMagmaInputString, 21  
CreateDirIfMissing, 19  
DifferencesList, 8  
distinct and common representatives, 10  
DistinctRepresentatives, 10  
EpimorphismByGenerators, 17  
ExponentOfPrime, 13  
FindMatchingFiles, 19  
Fitting series, 16  
FittingLength, 16  
FloatQuotientsList, 8  
GeneratorsAndInverses, 16  
GitHub repository, 4  
IntOrOnfinityToLaTeX, 20  
IsCommonTransversal, 10  
IsCommuting, 15  
LaTeXStringFactorsInt, 20  
ListOfPowers, 16  
Log2HTML, 19  
LowerFittingSeries, 16  
NextProbablyPrimeInt, 14  
OKtoReadFromUtils, 24  
PrimeNumbersIterator, 14  
PrintApplicableMethod, 20  
PrintOneItemPerLine, 6  
PrintSelection, 7  
QuotientsList, 8  
RandomCombination, 10  
RestrictedPartitionsWithout-  
    Repetitions, 13  
SearchCycle, 9  
SetIfMissing, 18  
smooth integer, 12  
StringDotSuffix, 11  
UpperFittingSeries, 16